
Глава 12. Методы
Мы использовали методы прежде и видели краткое обсуждение по их использованию в Главе 5. Здесь мы рассмотрим методы более углубленно. Методы являются такой основной частью программирования в CA-Visual Objects, что Вы поймете, что большинство вашего времени будет потрачено, программируя их. Эта глава будет более практическая, чем Глава 5 - мы имеем в своем распоряжении небольшое приложение, чтобы работать с некоторыми полезными советами по ходу изучения. В частности мы представим класс IniFileSpec и покажем, как его можно использовать и затем позже расширить.

Сначала, быстро напомним некоторые основы. Класс может иметь – 
· Переменные Экземпляра или Свойства - Переменные определялись как часть определения Класса. Каждый Экземпляр этого Класса (Объекта) имеет набор тех переменных, которые определяют свойства этого объекта.

· Методы - эффективные Функции определенные как часть Класса, с которым Вы можете управлять свойствами объектов того класса или исполнять некоторое другое действие, связанное с объектом класса.

Метод написанный для Класса может быть применен к любому объекту того класса для которого он был написан. Когда Вы используете SELF в Методе, то Вы ссылаетесь на Объект, рассматриваемый в данный момент.
Методы и свойства доступные объекту, автоматически включают методы и свойства, указанные непосредственно, включая СКРЫТЫЕ методы и свойства, и любые унаследованные методы и свойства; то есть любые нескрытые методы и свойства классов в цепочке наследования до рассматриваемого класса но не ниже него. Особенный метод может быть модифицирован внизу цепочки наследования, но лишь тот который вызывался через SELF - 'самая последняя версия'. Аналогично, действует то же имя метода с вызовами через SUPER ' предпоследняя версия ', но Вы не можете вызвать никакую более раннюю версию чем эта. См. Таблицу 12.1.
	Класс
	Метод написанный для класса
	Вызываемый через SELF
	Вызываемый через SUPER

	Class1
	HIDDEN Method1
Method2 (1)
	Method1
Method2 (1)
	

	Class2
	Method2 (2)

Method3
	Method2 (2)

Method3
	Method2 (1)



	Class3
	Method4
	Method2 (2)

Method3

Method4
	Method2 (1)

	· На каждом уровне, методы написаны для различных классов, но Method2(), написан как для Class1 так и Class2. В Class2 Вы имеете эффективную опцию, чтобы вызвать любую версию. Вызвать SELF:Method2(), что написанный в Class2 и вызвать SUPER:METHOD2() написанные в Class1. Вы не часто будите так делать, но это возможно.

· Method1() является СКРЫТЫМ только в  пределах кода, написанного в Class1. т.е. виден он будет только в коде написаном для Class1
· В Class3, доступны обе версии Method2(), хотя ни одна из них не была написана определенно для этого класса.
· Наконец, никакой класс не может видеть методы, написанные в классах ниже по цепочке наследования. Таким образом Method3() и Method4() - не доступены Class1.



SUPER не может применяться к переменным экземпляра - есть только одна версия каждой переменной экземпляра в объект того класса.

Метод - это просто функция, привязанная к классу, где синтаксис был изменен так, чтобы, вместо обращения к классу через параметр функции, его установили за пределами круглых скобок для большей выразительности. Понимание этого может помочь некоторым читателям понять как использовать методы. Методы были задуманы как новый вид функции для тесного использования в классах. Синтаксис, изобретенный для их использования прост и интуитивен.
Вы возможно подумаете, что длинная цепочка наследования должна сделать приложение медленным во время выполнения. Но это не так, потому что все наследование решено во времени компиляции.

Мы говорим о методах, 'посылаемых' объектам и ' : ', или Send, Оператор. При использовании метода, объект, которому он принадлежит, должен появиться в синтаксисе, поскольку много классов могут иметь подобно-названные методы (Polymorphism). Также, могут существовать много объектов того же самого класса. Связь создается цитированием Объектного Имени переменной, сопровождаемое Посылающим Оператором и Названием(именем) Метода как в -

<oObject>:<Method()>

Подобно функциям, методы должны всегда возвращать значение, даже если возвращенное значение не используется и игнорируется. Однако, если Вы опускаете инструкцию RETURN, заданное по умолчанию поведение должно возвратить SELF. Фактически, Init() методы никогда не могут возвращать что-нибудь другое, кроме SELF.

Пример Приложения 

Мы должны сформировать приложение, которое имеет ShellWindow без меню, метод Start(), запускающий DialogWindow, который позволяет пользователю ввести дату рождения непосредственно или через всплывающий Календарь. В зависимости от введенной даты и содержания INI файла, содержание двух Окон списка (ListBox) показывает состояние зрелости заинтересованного человека. Это окно также имеет кнопки (PushButtons), которые высвечивают на экране две версии текстового окна, один с содержимым INI файла и другого с мини файлом  Справочника.

После закрытия диалогового окна Adulthood Checker, Вы можете изменять размер, местоположение и состояние (Максимизировать/Минимизировать) ShellWindow и затем закрыть его. Когда потом его снова откроете, если файл INI не был удален или несанкционированно изменен, Оболочка будет сконфигурирована, так как она находилась перед предыдущем закрытии. Не очень полезно, но это хороший способ продемонстрировать, чего можно сделать с INI файлами.

Ch12a.AEF для этого приложения находится на CD-ROMЕ, но весь код цитируется ниже и мы рекомендуем, чтобы Вы работали с нами.
(Прим. пер.: Примеры к книге можно скачать с сайта http://ipavels.narod.ru в разделе «Файлы» )
Основной акцент главы должен объяснить использование методов.

В Проекте EVO, создайте новое приложение, называв его "Ch12a". Выберите Empty Application на вкладке Basic Прикладной Галереи, и отметьте пункты для Библиотек GUI и System. Переименуйте значение по умолчанию Module1 на "C:Start", откройте Редактора Исходного кода и введите следующий код –
CLASS DOBShell INHERIT ShellWindow

METHOD Start() CLASS App
LOCAL oShell AS DOBShell

LOCAL oWin AS DOBDataDialog
SetEpoch(Year(Today() - 32872))

SetCentury(TRUE)

SetDefault("C:\TEMP\")

oShell := DOBShell{SELF}

oShell:Caption := "Exploring CA-Visual Objects 2.5"

oShell:Show()

oWin := DOBDataDialog{oShell}

oWin:Show()

SELF:Exec()
RETURN NIL
Вышеупомянутый код делает некоторую подготовку перед созданием и открытием ShellWindow обычным способом. Сначала мы устанавливаем ' прокручиваем эпоху' которая в 1999, устанавливает эпоху до 1909 (сегодня минус 90 лет) так, чтобы дата данных не должна включать столетия. Если Вы введете 20/01/82, то программа предположит, что это 1982 год, а 1982 находится между 1909 и 2009. Для любого человека старшее чем 90 лет, год должен быть введен полностью. SetCentury(TRUE) здесь отображает 4-летние цифры, однако значение было введено. Вы позже увидите, что мы должны определить FieldSpec для даты и цитировать его для элемента управления в окне, которое должно быть окном DataDialog, так как FieldSpecs - не используется в DialogWindows. Это мог одинаково хорошо быть стандарт DataWindow.

Нет никаких файлов данных, но мы все же используем здесь SetDefault(), потому что это обеспечивает удобный способ определения пути для INI-файла и Вы можете очень хорошо решать, что ваш INI файл является файлом данных и должен быть сохранен другими данными. C:\Temp будет обязательно существовать (см. Главу 1) так что это кажется подходящим местоположением для нашего INI файла.

Затем мы создаем и открываем ShellWindow и DataDialogWindow, каждый из которых имеют код, который будет объяснен ниже. Мы приняли использование унаследованной версии ShellWindow, DOBShell и увидим почему, ниже.

Сохраните вашу работу и закройте редактор.

DOBDialogWindow
Создайте новый модуль " W:DOBDialogWindow " и откройте Редактор FieldSpec. Создайте FieldSpec с именем "DOB"; мы используем его вскоре в Редакторе Окна. Укажите обоим FieldSpec, свойства Name как "DOB", Type установите как <Date> и Required в <No> (Рис. 12.1). Сохраните FieldSpec и закройте Редактор FieldSpec.

	Рис 12.1 

В Редакторе FieldSpec, выберите свойствам FieldSpec и Name "DOB", установите свойству Type Date  и Required в No. 


Откомпилируйте приложение так, чтобы ресурс FieldSpec был доступен Редактору Окна. Браузер Ошибок сообщит Structure or Class DOBDataDialog unknown (Структура или Класс DOBDataDialog неизвестна). Достаточно справедливо - мы все же не создали его; давайте сделаем это сейчас. Закройте Браузер Ошибок.
В Редакторе Окна создайте DataDialog по имени "DOBDataDialog" с Caption  "Adulthood Checker" и установите его размеры 400x240. 
Положите на него следующие элементы управления –

· Элемент управления FixedText в позицию 42,33 и размером 72x20. Установите свойства Caption в "Date of Birth:" и Generate Code в No.

· Элемент управления FixedText в позицию 29,84 и размером 80x20. Установите свойство Caption в "Can Do" и свойство Generate Code в No.

· Элемент управления FixedText в поз. 141,84 и размер 80x20. Установите свойство Caption в "Can’t Do" и cвойство Generate Code в No.
· Элемент управления FixedText в 291,33 и размер 80x20. Установите свойство Caption в "Age:" и свойство Name в "ftFuture".

· Элемент управления SingleLineEdit в 117,33 и размер 80x20. Установите свойство Name в "sleDOB", свойство Description в " Date of Birth " и свойство FieldSpec в DOB.

· Элемент упралвения ListBox в 29,107 и размер 80x100. Установите cвойство Name в "lbCanDo" и cвойство Description в "Contains legal activities".

· Элемент управления ListBox в 141,107 и размер 80x100. Установите свойство Name в "lbCantDo" и свойство Description в "Contains illegal activities".

· Элемент управления PushButton в 231,27 и размер 32x32. Установите свойство Name в "pbCalendar" и свойство Description в " Display calendar for alternative method of date entry ".

· Элемент управления PushButton в 271,108 и размер 80x20. Установите свойство Caption в "&Help", свойство Name в "pbHelp" и свойству Description "Show a Description of this application".
· Элемент управления PushButton в 271,142 и размер 80x20. Установите свойство Caption в "&Edit INI ", свойство Name в "pbINI" и свойству Description  "Display the INI file text".
· Элемент управления PushButton в 271,176 и размер 80x20. Установите свойство Caption в "&Cancel", свойство Name в "pbCancel" и свойство Description в "Close the Dialog". Сохраните Окно, перейдите на вкладку General и щелкните по свойству Click Event, чтобы открыть Редактора Исходного кода  вместе с методом pbCancel() на первой линии. Отредактируйте код –
METHOD pbCancel( ) CLASS DOBDataDialog

SELF:EndWindow()

RETURN NIL
Сохраните метод pbCancel(), закройте Редактор исходного кода, сохраните окно и закройте Редактор Окна. Откомпилируйте и выполните приложение; Окно Adulthood Checker (Рис. 12.2) должно быть отображено, если все прошло так как планировалось. За исключением кнопки Cancel, ничто другое пока не работает.
Обратите внимание, что маленький квадрат элемента управления PushButton все еще показывает значение свойства Caption по умолчанию Push. Мы хотим, чтобы на ней было изображение, но мы еще не видели, как создавать ресурс Bitmap (об этом будет рассказано в Главе 20). Однако, модуль изображения есть на CD-ROMе. Нажмите Cancel и закройте окно приложения. Копируйте файл Calendar.MEF в каталог C:\EVO\Ch12 из каталога Ch12a на КОМПАКТ-ДИСКЕ и затем импортируйте файл с вашего жесткого диска. Рис. 12.3 показывает точечный рисунок в Редакторе Изображения.

	Рис 12.2

Окно Adulthood Checker, ни чего пока не функционирует


 Откомпилируйте приложение так, чтобы ресурс изображения был доступен Редактору Окна. Теперь откройте DOBDataDialog в Редакторе Окна и выберите Calendar как свойство Image на вкладке General для  кнопки pbCalendar. Это заставит изображение появится, как на Рис. 12.4, Вы увидите после сохранения вашей работы, компиляции и запуска приложения. Уровень детали, показанной в Редакторе Изображения весьма достаточен, когда изображение уменьшено до размера, необходимого кнопки.

	Рис 12.3
Это точечный рисунок для кнопки pbCalendar созданный Редакторе изображения


Теперь, когда мы получили основную структуру нашего приложения, мы обратим наше внимание к сохранению размера, позиции и состояния Окна Приложения (ShellWindow) в INI файл для того, чтобы, когда затем откроем Окно Приложения вновь, то оно появится точно в такого же размере и позиции, когда оно было закрыта. В то же самое время мы будем использовать INI файл, чтобы хранить данные справочников, которые будут использоваться позже алгоритмом проверки возраста.

INI Файлы и Класс, чтобы управлять ими

Откройте метод Start() и измените  его на это -

METHOD Start() CLASS App

LOCAL oShell AS DOBShell

LOCAL oWin AS DOBDataDialog

LOCAL aTemp AS ARRAY // New Line
SetEpoch(Year(Today() - 32872))

SetCentury(TRUE)
SetDefault("C:\TEMP\")
// New section ...
oINI := INIFileSpec{GetDefault() + "DOBINI.INI"}
IF ! oINI:lFileExists

aTemp := {    {"Drinking","18"}, ;

{"Marrying", "16"}, ;

{"Voting", "18"}, ;

{"Driving", "17"}}
oINI:WriteSection("Min Age", aTemp)
aTemp := {    {"XOrigin", "100"}, ;

{"YOrigin", "150"}, ;

{"Width", "500"}, ;

{"Height", "350"}, ;

{"ShowState", "0"}}
oINI:WriteSection("Shell Define", aTemp)

ENDIF
// End of new section
oShell := DOBShell{SELF}

oShell:Caption := "Exploring CA-Visual Objects 2.5"

oShell:Show()
oWin := DOBDataDialog{oShell}

oWin:Show()

SELF:Exec()

RETURN NIL
Оставьте пустую строку и добавьте следующее объявление –
GLOBAL oINI AS INIFileSpec
Сохраните код и закройте редактора.

	Рис 12.4
Когда рисунок уменьшен до размера кнопки, работа становится приятнее 


<oINI> - глобальная переменная, используется чтобы содержать ссылку на объект класса IniFileSpec (который будет описан вскоре). Однако, если никакого INI файл не существует для нашего приложения, мы должны создать один со значениями по умолчанию. Мы увидем ниже, как работает метод WriteSection(), но Вы уже можете  видеть, как он эффективно записывает из массива два столбца в INI файл и он используется как упоминалось выше, чтобы сохранить значения по умолчанию для проверки совершеннолетия, а так же размер и начало координат Окна Приложения.
INI файлы были представлены с 16-разрядными виндами для параметров настройки конфигурации. Они могут использоваться для многих вещей, включая маленькие количества данных, которые не соответствуют формату DBF с многочисленными, аналогичными записями. Microsoft же рекомендует, чтобы мы использовали Системный реестр в 32-разрядных приложениях.

Для целей этой вводной книги, мы полагаем, что более безопасно объяснить использование INI файлов, так как они могут использоваться аналогичным способом как системный реестр. Многое из кода, описанного здесь для записи и чтения в/из INI файла могло быть непосредственно заменено кодом, чтобы записать и читать в/из системного реестра, в другое время.

Однако, есть случаи использования INI файлов для данных неконфигурации даже в приложениях, где системный реестр используется для параметров настройки конфигурации. Так что это - не потраченное впустую усилие.
INI файлы - текстовые файлы со специфическим форматом. Разделы имеют Заголовки и каждый раздел может иметь множество Данных. Заданное по умолчанию содержимое нашего INI файла было установлено в коде выше и вот его содержимое –
[Min Age]

Drinking=18

Marrying=16

Voting=18

Driving=17
[Shell Define]

XOrigin=100

YOrigin=150

Width=450

Height=300

ShowState=0

Как Вы видите, заголовки раздела заключены в квадратные скобки и каждые данные  состоят из одиночной линии текста, который всегда включает знаки "=". Текст слева от знака "=" – Ключ, а что справа является Значением. Ключи должны быть по крайней мере длиной в два символа.  В нашем случае, все значения Ключа, это числа, но фактически они являются текстом и нуждаются в преобразовании для использования их как числа. Промежутки могут быть оставлены с любой стороны от знака "=", но обычно опускаются. Хотя ключи всегда сохраняются, как написано они - не зависимы от регистра.
Включением точек с запятой, обозначают комментарии, как тут –

[User]


  ; Comments are not allowed on section

        ; name lines

UserName=Alan Jones      ; but are allowed on their own or as here

Мы будем должны использовать запросы функции (Application programming interface) Windows API работая с INI файлами. Windows API является опубликованным набором функций, которые используются 32-разрядными операционными системами Windows (NT и '9x) и доступный всем приложениям, написанные для выполненния на этих операционных системах. Поэтому мы полагаем что PC пользователя имеет к ним доступ. CA-Visual Objects использует эти функции для полного взаимодействия с объектами окна и операционной системой, но переносит запросы к этим функциям в пределах классов CA-Visual Objects или его собственных функций. Мы сделаем наш собственный перенос для обработки INI файла, потому что она не существует как особенность CA-Visual Objects.

Вы обратите внимание, что Вы можете вызывать функции WinAPI непосредственно,  потому что они были разработаны для использования с языком Си, они расширяют использование строк с нулевым окончанием (PSZs) и тип данных Структуры. Мы будем использовать их, но рассматривать полное объяснение их использования вне контекста этой книги.

Используя вызовы Windows API, мы можем читать или записывать индивидуальный ключ или целую секцию, или находить все названия секции и т.д. Наш класс IniFileSpec предназначен, чтобы делать эти процессы очень простыми. <oINI> был объявлен как ГЛОБАЛЬНАЯ ПЕРЕМЕННАЯ в модуле C:Start, а так же ему был создан и назначен объект IniFileSpec в методе Start(), чтобы он был всегда доступен по всему приложению.

Давайте начнем с создания класса IniFileSpec. Создайте новый модуль с названием "C:IniFileSpec", выберите этот модуль, откройте Редактора Исходного кода и введите следующий код –
CLASS INIFileSpec INHERIT FileSpec

PROTECT lFileExists AS LOGIC
DECLARE METHOD GetInt

DECLARE METHOD GetSection
ACCESS lFileExists CLASS INIFileSpec

RETURN SELF:lFileExists
METHOD Init(cFullPath) CLASS INIFileSpec

SUPER:Init(cFullPath)
IF ! SELF:Extension == ".INI"

SELF:Extension := ".INI"

ELSE

IF Empty(SELF:Drive) .AND. Empty(SELF:Path)

SELF:FullPath := DiskName() + ":\" + CurDir() ;

+ "\" + SELF:FullPath

ENDIF

IF File(SELF:FullPath)

SELF:lFileExists := TRUE

ENDIF

ENDIF
RETURN SELF
Сохраните ваш код. Это - хороший пример наследования. Мы используем Класс FileSpec и строим на нем, добавляя <lFileExists> и множество методов, которые должны быть написаны.

Мы уже обсудили методы ACCESS и ASSIGN, используемые с PROTECTed и HIDDEN переменными как средства достижения более устойчивого решения (Глава 5). ACCESS необходим для <lFileExists> в классе IniFileSpec, потому что мы решили сделать <lFileExists> PROTECTed (защищенным).

Мы также обсудили функции со строгим контролем типов (Глава 8). Строгий контроль типов стоит рекомендовать для методов по точно тем же самым причинам что касается функций со строгим контролем типов - Вы получите меньшее количество отказов во время выполнения и код во время выполнения будет работать быстрее. 

Метод Init () берет строковое значение, в нашем случае "C:\TEMP\DOBINI.INI". Запрос SUPER:Init () заполнит объявленные переменные FileSpec, и следовательно IniFileSpec, так что –
SELF:Drive = "C:"

SELF:Path = "\TEMP\"

SELF:FileName = "DOBINI"

SELF:Extension = ".INI"

Если бы, передали "DOBINI.INN", тогда конечный результат, была бы ссылка на файл DOBINI.INI (исправленное расширение) в текущем каталоге текущего диска.

Обратите внимание, что ни какой потребности в расширении для INI файла в этой точки не существует. Метод Init() проверяет, существует ли он и определяет <SELF:lFileExists> соответственно. Запись в файл автоматически создает его, в случае необходимости и изменяет значение его свойства на True. Мы теперь имеем в своем распоряжении Объект IniFileSpec, на котором создание и вся связь с INI файлом будут делегированы.

Перед рассмотрением более сложного случая записи целого раздела в DOBINI файл, как было сделано в методе Start(), давайте, разберем строку программы, которая будет использоваться позже в приложении и увидим, как она работает –

oINI:WriteInt("Shell Define", "Width", SELF:Size:Width)

Здесь мы пишем числовое значение в INI файл. Эта инструкция фактически открывает файл, если он существует или создает его, если его несуществует. Затем  он находит или создает раздел по имени "Shell Define" и записывает "Width =350", или безотносительные значения, в него и затем закрывает файл. Так, чтобы записать данные в файл DOBINI.INI, мы должны делать так. Метод WriteInt(), который Вы должны добавить в модуль C:IniFileSpec, выглядит похожим на это –
METHOD WriteInt(cSection, cEntry, dwINT) CLASS IniFileSpec

SELF:lFileExists := ;

WritePrivateProfileString(PSZ(cSection), ;

PSZ(cEntry), PSZ(AllTrim(Str(dwINT))), ;

PSZ(SELF:FullPath))

RETURN SELF:lFileExists

Вторая строка - запрос к функции Windows API WritePrivateProfileString(), в которой мы должны конвертировать(преобразовать) параметры из STRING и DWORD типов данных в PSZ тип данных. Функция возвращает True,  в случае успеха, в этом случае INI файл теперь существует, даже если он не был перед этим создан и мы используют его, чтобы установить свойство <lFileExists>. Это значение, также возвращено из метода в случае, если мы хотим проверить наличие успеха в вызываемом коде.
Наше беспокойство - не так много, чтобы объяснить запрос к функции WinAPI как использование метода непосредственно. WinAPI функции могут усложняться и это  вне контекста этой книги. Этот набор Функций API настолько полезен, тем не менее,  они имеют смысл копируясь со своим классом суперобертки. Если у вас есть книга по Windows API, то  у вас не будете много проблем, используя много других функций WinAPI, особенно если Вы имеете опыт программирования на Си.

Оглядываясь назад, метод Start() нашего приложения, записывает следующий код  целого раздела [Min Age], в DOBINI.INI –

aTemp := {  {”Drinking”, “18”}, ;

{”Marrying”,” 16”}, ;

{”Voting”,”18”}, ;

{”Driving”, “17”}}

oINI:WriteSection(“Min Age”, aTemp)

Метод WriteSection() использует функцию WinAPI WritePrivateProfileSection() и должен также быть скопирован в модуль C:IniFileSpec –
METHOD WriteSection(cSection, aEntry) CLASS IniFileSpec

LOCAL cString AS STRING

LOCAL dwCounter AS DWORD
cString := ""

FOR dwCounter := 1 UPTO ALen(aEntry)

cString += AllTrim(aEntry[dwCounter, 1]) ;

+ "=" + AllTrim(aEntry[dwCounter, 2]) + _Chr(0)

NEXT

SELF:lFileExists := ;

WritePrivateProfileSection(PSZ(cSection), ;

PSZ(cString), PSZ(SELF:FullPath))
RETURN SELF:lFileExists
Вы увидите, что этот код формирует одну строку из массива и передает ее к функции WinAPI, каждая часть строки, заканчиваемой нулем в соответствии с PSZ() тип данных, т.е. символом _Chr(0). Мы не будем углублятся в содержание всех методов класса IniFileSpec, так как они доступны на КОМПАКТ-ДИСКЕ в файле IniFileSpec.AEF. Однако, нам потребуется примеры следующих двух методов в нашем небольшом приложении, так что они должны также быть введены в C:IniFileSpec –
METHOD GetInt( cSection AS STRING, cEntry AS STRING) ;

AS INT CLASS IniFileSpec

LOCAL liRet AS INT
IF SELF:lFileExists

liRet := GetPrivateProfileInt(PSZ(cSection), ;

PSZ(cEntry), 0, PSZ(SELF:FullPath))

ENDIF
RETURN liRet
Вышеупомянутый метод, конечно, преобразованный из описанного выше WriteInt().

METHOD GetSection(cSection AS STRING) AS ARRAY

CLASS IniFileSpec

LOCAL dwBytes, dwEnd, dwPos AS DWORD

LOCAL cString, cEntry AS STRING

LOCAL aEntry AS ARRAY
aEntry := {}

IF SELF:lFileExists

cString := Space(2500)

dwBytes := GetPrivateProfileSection(PSZ(cSection), ;

PSZ(cString), SLen(cString), ;

PSZ(SELF:FullPath))

cString := Left(cString, dwBytes)

DO WHILE(dwEnd := At3(_CHR(0), cString, 2)) > 0

cEntry := Left(cString, dwEnd - 1)

AAdd(aEntry, cEntry)

cString := SubStr2(cString, dwEnd + 1)

ENDDO

AEvalA(aEntry, {|e| dwPos := At2("=", e), ;

{SubStr3(e, 1, dwPos - 1), ;

SubStr2(e, dwPos + 1)}})

ENDIF
RETURN aEntry

Точно так же GetSection() преобразован из WriteSection(), который был описан выше. <aEntry> создан, как пустой массив, а <cString> установлен как буфер длиною 2500. Она передается к Функции API, которая изменяет содержание <cString> и возвращает длину <dwBytes>, который определяет длину полезной информации, помещенной в <cString>. Дополнительная длина <cString> отброшена и полезная часть разбита на отдельные строки, основанные после обнаружения значения _Chr(0) в границах каждого данного, в пределах <cString>. Поскольку каждая данная  в разделе найден, она добавляется как новый элемент к массиву <aEntry>. Функция AEvalA() потом оценивает блок кода, чтобы делить каждый элемент в пределах <aEntry> на две части: Ключ и Значение, которые помещаются в подмассив. Таким образом, Voting=18 становится {'Voting', 18}. Блоки кода будет объясняться в следующей главе.

Перед продолжением, мы предлагаем, вам чтобы Вы сохранили и откомпилировали вашу работу, чтобы гарантировать, что все еще идет хорошо.

Теперь, когда мы имеем в своем распоряжении класс IniFileSpec, мы можем обратить наше внимание на использование его, чтобы отобразить ShellWindow с началом координат и размерами как определено в INI файле.

Окно Приложения
Мы должны использовать унаследованную версию класса ShellWindow, потому что нам нужно включить код в внутри него, чтобы извлечь значения из INI файла, чтобы определить начало координат и размер Окна Приложения. Мы можем извлечь эту информацию внутри метода PostInit(), но мы решили сделать это в методе Show().

Введите следующий код в модуль C:Start –

METHOD Show(uShowState) CLASS DOBShell
LOCAL liWidth, liHeight, liX, liY AS INT
IF oINI:lFileExists

liX := oINI:GetInt("Shell Define", "XOrigin")

liY := oINI:GetInt("Shell Define", "YOrigin")

liWidth := oINI:GetInt("Shell Define", "Width")

liHeight := oINI:GetInt("Shell Define", "Height")
IF liHeight > 0

SELF:Size := Dimension{liWidth, liHeight}

SELF:Origin := Point{liX, liY}

ENDIF
IF IsNil(uShowState )

SUPER:Show(oINI:GetInt("Shell Define", "ShowState"))

ELSE

SUPER:Show(uShowState)

ENDIF
ELSE

SUPER:Show(uShowState)

ENDIF
RETURN NIL
В метде DOBShell:Show() мы читаем значения из  INI файла, затем устанавливаем начало координат и размеры окна перед вызовом SUPER:Show(), чтобы отобразить его.

При закрытии приложения мы должны сохранить текущие координаты и размер Окна Приложения в DOBINI.INI. Эти значения больше не будут доступны после того, как Окно Приложения закроется, поэтому они должны быть сохранены в течение закрытия этого окна. Запомните, что один из Методов Callback  всех окон – метод Close() и мы сохраним координатs и размер Окна Приложения в нем. Если мы создадим наш собственный метод DOBShell:Close(), снова вызывав SUPER:Close() чтобы получить нормальное поведение, мы сможем вставить наш собственный код, чтобы сохранить состояние окна приложения в DOBINI.INI. Снова, следующий код должен быть введен в C:Start –
METHOD Close(oE) CLASS DOBShell

SELF:SaveState()

SUPER:Close(oE)

RETURN NIL
В вышеупомянутом методе, запрос к SaveState() происходит прежде, чем  SUPER:Close(oE), чтобы гарантировать, что требуемые значения все еще доступны, чтобы сохраненить их.

Наконец, в модуль C:Start по крайней мере, введите следующий код –

METHOD SaveState() CLASS DOBShell
LOCAL liShowState AS INT
IF SELF:IsZoomed()

liShowState := SHOWZOOMED

oINI:WriteInt("Shell Define", "ShowState", liShowState)

ELSEIF SELF:IsIconic()

liShowState := SHOWICONIZED

oINI:WriteInt("Shell Define", "ShowState", liShowState)

ELSE

liShowState := SHOWNORMAL

oINI:WriteInt("Shell Define", "XOrigin", ;

iif(liShowState = SHOWNORMAL, SELF:Origin:X, 0))

oINI:WriteInt("Shell Define", "YOrigin", ;

iif(liShowState = SHOWNORMAL, SELF:Origin:Y, 0))

oINI:WriteInt("Shell Define", "Width", SELF:Size:Width)

oINI:WriteInt("Shell Define", "Height", SELF:Size:Height)

oINI:WriteInt("Shell Define", "ShowState", liShowState)

ENDIF

oINI := NULL_OBJECT

RETURN NIL

Здесь объект IniFileSpec, <oINI>, используется, чтобы записать назад в DOBINI и затем объект завершен и может быть установлен в NULL_OBJECT. Координаты и размер окна приложения сохранены, только если знаечение <liShowState> равно SHOWNORMAL. Это способ, если Окна Приложение было закрыто в то время когда максимизировалось или минимизировалось, при повторном запуске приложения, восстановление Окна Приложения, возвратит это к размеру и позиционирует его, прежде чем имело место максимизирование или минимизирование.

Откомпилируйте и запустите приложение. Оно выполнится как прежде, но на сей раз, до закрытия диалогового окна Adulthood Checker, измените размер и/или позицию Окна Приложеня перед его закрытием. Запустите приложение снова и Вы увидите, что Окно Приложения появится на экране точно таким же, каким оно находилось до закрытия. Когда Вы закончите, проверяя его, закройте приложение и, если Вы желаете, вы можите открыть INI-файл в текстовом редакторе, чтобы проверить его содержание.

Предоставление Adulthood Checker
Данные, используемые в диалоговом окне Adulthood Checker те, что доступны в секции [Min Age]  файла DOBINI.INI. Она должна сравниватся с возрастом, даты рождения которое введено в элемент управление SingleLineEdit. Поэтомы мы должны, получить значения из INI файла. Ее имеет смысл сохранить в массив для сравнений в коде и для этого, мы нуждаемся в новом свойстве окна. Откройте окно DOBDataDialog в Редакторе Окна и на вкладке DataDialog, выберите свойство Class Declaration. Добавьте следующую переменную после строки //{{%UC %}} USER CODE... –
//{{%UC%}} USER CODE STARTS HERE (do NOT remove this line)
PROTECT aLookup AS ARRAY

Это дает нам место, чтобы сохранить значения. Затем, создайте метод PostInit() для окна, щелкнув по свойству PostInit и изменяя его на следующее –
METHOD PostInit(oWindow, iCtlID, oServer, uExtra) ;

CLASS DOBDataDialog

//Put your PostInit additions here

SELF:aLookup := oINI:GetSection("Min Age") // New Line

RETURN NIL

Методы PostInit() использовались прежде и описаны более подробно ниже. В этом случае, содержание < SELF:aLookup > будет –
{
{”Drinking”,
18},

{”Marrying”,
16},

{”Voting”,
18},

{”Driving”,
17} }
Теперь мы должны изменить содержимое Окона списка (ListBox) в соответствии с законами возраста и ссылатся на дату рождения. Для этого, мы можем использовать метод EditFocusChange() окна. Перейдите на вкладку Control Events, выберите свойство EditFocusChange и измените код на –
METHOD EditFocusChange(oEditFocusChangeEvent) ;

CLASS DOBDataDialog

LOCAL oControl AS Control

LOCAL lGotFocus AS LOGIC
oControl := IIf(oEditFocusChangeEvent == NULL_OBJECT, ;

NULL_OBJECT, ;

oEditFocusChangeEvent:Control)

lGotFocus := IIf(oEditFocusChangeEvent == NULL_OBJECT, ;

FALSE, ;

oEditFocusChangeEvent:GotFocus)

SUPER:EditFocusChange(oEditFocusChangeEvent)

//Put your changes here

IF oControl = SELF:oDCsleDOB .AND. ! lGotFocus

IF ! IsNil(sleDOB)

SELF:FillLists()

ENDIF

ELSEIF oControl = SELF:oDCsleDOB .AND. lGotFocus

SELF:oDCftFuture:TextValue := ""

ENDIF
RETURN NIL

Это - Метод Повторного вызова, о котором чуть больше ниже. Он вызывается дважды, когда фокус переходит от одного элемента редактирования к другому: первый раз, при оставлении первого элемента управления и второй раз, при входе на следующий. <oControl> - элемент управления, оставляемый или введенный и <lGotFocus> True при входе в <oControl>. Все это установлено для Вас в сгенерированном коде выше строки  //Put your changes here. 
Код, для заполнения Окна списка содержится в отдельном методе, потому что мы также могли вызвать его где-нибудь. Добавьте этот код к модулю W:DOBDialogWindow –
METHOD FillLists() CLASS DOBDataDialog
IF SELF:sleDOB > Today()

SELF:oDCftFuture:TextValue := "Reincarnation?!"

SELF:oDClbCanDo:Fillusing({})

SELF:oDClbCantDo:Fillusing({})

ELSE

SELF:oDCftFuture:TextValue := ""

SELF:oDClbCanDo:Fillusing(SELF:CanDo(TRUE))

SELF:oDClbCantDo:Fillusing(SELF:CanDo(FALSE))

ENDIF
RETURN NIL

Мы заинтересованы только при оставлении Date of Birth SingleLineEdit или Calendar PushButton (еще не включенный) или при повторном входе SingleLineEdit. В вышеупомянутом коде, все другие элементы управления игнорируются.

<sleDOB> хранит значение, введенное в SingleLineEdit Date of Birth (ACCESS, сгенерированный Редактором Окна). Оно содержит или НОЛЬ или значение Date. Если НОЛЬ, она пока игнорируется, но если она содержит дату, значение - по сравнению с сегодняшней датой. Если дата находится в будущем, Окна списка заполнены пустыми массивами. Иначе, Окна списка заполнены как соответствующий.

Эти два Окна списка, CanDo и CantDo, заполнены в EditFocusChange() выше использования метода ListBox:FillUsing() Метод согласно тому, является ли <lCan> TRUE нет. Методы FillUsing() используют возвращаемое значение (массив) от метода CanDo(), цитируемый ниже. Этот метод не знает название Редактора Окна, так что он должен быть создано вручную в модуле W: DOBDataDialog Module-
METHOD CanDo(lCan) CLASS DOBDataDialog
LOCAL aFill AS ARRAY

LOCAL dwCounter AS DWORD
aFill := {}
FOR dwCounter := 1 UPTO ALen(SELF:aLookup)

IF lCan

IF ! SELF:IsUnderAge(SELF:aLookup[dwCounter, 1])

AAdd(aFill, SELF:aLookup[dwCounter, 1])

ENDIF

ELSE

IF SELF:IsUnderAge(SELF:aLookup[dwCounter, 1])

AAdd(aFill, SELF:aLookup[dwCounter, 1])

ENDIF

ENDIF

NEXT
RETURN aFill
Вышеупомянутый метод создает пустой массив <aFill> и добавляет элементы к нему вызывая функцию AAdd(). Значение по умолчанию INI файл имеет четыре данных, так что Alen(SELF:aLookup) возвращается 4. Если <lCan> TRUE и SELF:IsUnderAge() возвращает FALSE, элемент добавлен к <aFill>. Содержание SELF:aLookup(dwCounter, 1) может быть 'Voting', когда она добавлена в ListBox, определенному для <lCan> TRUE. Метод CanDo() поэтому возвращает массив значений, чтобы загрузить в соответсвующий ListBox, при сравнивнении возраста вычисленного из даты рождения с совершеннолетним возврастом для деятельности в методе IsUnderAge(), перечисленный ниже, которого должен быть вручную введен снова, подобно предыдущему методу в модуль W: DOBDialogWindow –
METHOD IsUnderAge(cActy) CLASS DOBDataDialog
LOCAL uRet AS USUAL

LOCAL dwAge AS DWORD

LOCAL uMinAge AS USUAL
dwAge := Age(SELF:sleDOB, Today())

SELF:oDCftFuture:TextValue := "Age: " + NTrim(dwAge)

uMinAge := ATranslate(SELF:aLookUp, cActy)

IF IsNil(uMinAge)

uRet := NIL

ELSE

uRet := dwAge < Val(uMinAge)

ENDIF
RETURN uRet
Вышеупомянутый метод использует функцию Age(), чтобы вычислить возраст, беря дата рождения и сегодняшнию дату, как описано ниже. Метод возвращает TRUE, FALSE, или NIL. Если INI файл не фальсифицирован, этот метод никогда не должен возвратить NIL. Мы также разместим функцию Age() в модуле W: DOBDialogWindow –
FUNCTION Age(dDate1 AS DATE, dDate2 AS DATE) AS DWORD

LOCAL dwAge AS DWORD
dwAge := Year(dDate2) - Year(dDate1)

dwAge -= IIf(Month(dDate2) * 32 + Day(dDate2) ;

< Month(dDate1) * 32 + Day(dDate1), 1, 0)
RETURN dwAge
Почему использовали функцию здесь? Серьезное основание - то, что этот код может быть использован полностью не имеющим отношения клюдям и быть не совершеннолетними, так связь кода к классу ограничина. В строке 4, используется множитель 32, являющийся больше чем число дней в любом месяце - OK, это выглядит забавным, но это работает!

Мы поместили вышеупомянутую функцию в тот же самый модуль как предварительно введенные методы. Вы возможно предпочитате хранить функции в своем собственном модуле, поскольку они имеют большее количество общей применимости.

Окомпилируйте и запустите приложение. Теперь, Вы можете ввести Date of Birth (Дату рождения) в SingleLineEdit и, когда Вы потом нажмете <Tab>, вызовится метод  EditFocusChange(), вызывая на заполнения Окна списка соответствующими данными (Рис. 12.5).

	Рис 12.5

Окно Adulthood Checker, показывая действиям, которые не может  делать 16-летний.


Закройте приложение для того, чтобы мы могли создать Окно Calendar, которое позволит пользователям выбирать дату рождения в элементе управления Month Calendar.
Окно Календаря
Чтобы кнопка pbCalendar могла открывать окно, содержащее элемент управление Calendar, мы должны создать для этого метод. Откройте DOBDataDialog в Редакторе Окна, выберите кнопку pbCalendar и на вкладке General, щелкните по свойству  Click Event; измените текст следующим образом –
METHOD pbCalendar( ) CLASS DOBDataDialog

LOCAL oWin AS CalendarWin

oWin := CalendarWin{SELF, SELF:sleDOB}

oWin:Show()

RETURN NIL
Сохраните и закройте все редакторы.

Теперь создайте новый модуль "W:Calendar" с новым объектом, в котором Вы должны создать Диалоговое Окно (DialogWindow) с именем "CalendarWin". Измените свойство Caption, на "Select Date of Birth" и устновите ширину окна 230, а высоту 250. Разместите элемент управления Month Calendar в позиции 25,30 и элемент управления PushButton в позиции 75,210. Измениет свойство Caption PushButton на "OK" и Name  на "pbOK". (Рис. 12.6)
Элемент управление Calendar позволяет пользователю просматривать дату или устанавливать диапазон даты. Годы изменяются, нажатием непосредственно на годы, который потом отображает элемент управления Spinner. Щелчок в месяце отображает меню месяцев, чтобы выбрать или же Вы можете выбрать месяц, щелкая по кнопкам слева и справо. Кнопка OK будет содержать код, чтобы передать выбранную дату окну-владельцу и закрытие Диалога. Сохраните окно.

	Рис 12.6

Элемент управления Календаря Месяца применяется для того, чтобы пользователи могли выбирать требуемую дату, он возвращает результат в Date Birth, когда нажата OK.


Нам нужен метод PostInit() для CalendarWin, чтобы загрузить календарь с датой, если таковые вообще имеются, из элемента управления SingleLineEdit в окне-владелец. Вы увидите, что дата проходила в методе Init() в методе DOBDataDialog:pbCalendar() выше. Выберите CalendarWin и на вкладке DlgWindow, нажмите на свойство PostInit  Actions. Измените код на –
METHOD PostInit(oParent, uExtra) CLASS CalendarWin
//Put your PostInit additions here

SELF:oDCMonthCalendar1:Selection := uExtra // New Line
RETURN NIL
Вышеупомянутый код будет изменен ниже, но он передает дату из <uExtra> в элемент. Аналогично, когда метод pbOK() вызывается, выбор приходит назад к элементу управлению SingleLineEdit. Код для метода pbOK() должен также быть введен –
METHOD pbOK( ) CLASS CalendarWin

SELF:Owner:sleDOB := SELF:oDCMonthCalendar1:Selection

SELF:EndDialog(0)

RETURN NIL
Однако, при потере фокуса кнопки Calendar, метод EditFocusChange() не вызывается, потому что мы не вводим, ни оставляем средства редактирования. Мы должны вызвать FillLists() непосредственно после возвращения из Диалога Calendar и чтобы  сделать это, добавляем следующую строку кода в конец pbCalendar() в W:DOBDialogWin –
METHOD pbCalendar( ) CLASS DOBDataDialog
LOCAL oWin AS CalendarWin
oWin := CalendarWin{SELF, SELF:sleDOB}

oWin:Show()

SELF:FillLists() // New Line
RETURN NIL
Откомпилируйте и запустите приложение и нажмите на кнопку Calendar.
Что случилось? Как показано на Рис. 12.7, Ошибка Data Type сообщает, что был получен NIL, когда требуется тип данных Date. Объясняем - то, что CalendarWin вызвался из метода pbCalendar(), передавая <SELF:sleDOB> как второй параметр. Но, поскольку Calendar SingleLineEdit был пуст, <SELF:sleDOB> содержит NIL вместо ожидаемого Date.

	Рис 12.7

Сообщена Ошибка Типа данных, поскольку мы передали НОЛЬ элементу управлению Календарь вместо требуемого типа данных Date.


Имейте в виду, что, если метод - не со строгим контролем типов, параметры могут быть законно опущены при вызове метода. В таких случаях, включается в пределах метода средства установки значений по умолчанию для опущенных параметров, которые иначе приняли бы значение NIL. Это может быть достигнуто, используя функцию Default(), чтобы установить его значение в NULL_Date, если она содержит NIL. Устанавливая его в NULL_Date, значение по умолчанию устанавливает элемент управления Month Calendar в сегодняшную дате от системных часов.
Щелкните по Abort, чтобы закрыть Диалог Error и затем измените метод PostInit()  для CalendarWin следующим образом –
METHOD PostInit(oParent,uExtra) CLASS CalendarWin
//Put your PostInit additions here

Default(@uExtra, NULL_DATE) // New line

SELF:oDCMonthCalendar1:Selection := uExtra
RETURN NIL
Обратите внимание на синтаксис функции Default(). Передаем переменную по ссылке и если она равно Nil, приобретает значение, указанное во втором параметре.

Как с функциями со строгим контролем типов, методы со строгим контролем типов полностью решены в во время компиляции. Это очень улучшает проверку ошибок во время компиляции и скорость во время выполнения и эффективность использования памяти.

Единственная дополнительная вещь, которую надо помнить при использовании методов со строгим контролем типов в противоположность функциям со строгим контролем типов, состоит в том, чтобы объявить те имена метода в объявлении класса. Компилятор вынудит Вас сделать из-за  сообщений  об ошибках компиляции, так что нет большого бремени, чтобы запоминить не забыть сделать это.

Теперь, когда Вы откомпилируете и запустите приложение, Вы можите использовать кнопку Calendar, чтобы открыть Диалог CalendarWin для, того чтобы Вы могли выбрать Date of Birth в элементе управления Calendar значение, которое передается назад в SingleLineEdit, когда Вы нажимете на кнопку OK .

Редактор INI файла и средство просмотра справки

Ни одно из этих окон вероятно, не быдет готово в реальном приложении; справку лучше обеспечить через надлежащий справочный файл. Тем не менее, мы можем проиллюстрировать здесь несколько пунктов.
Это показывает другой пример наследования, потому что оба окна происходят от того же самого класса TextViewer DialogWindow. Оба имеют то же формат, но редактируемы, а другой нет. Свойство ReadOnly элемента управления как установлено в Редакторе Окна обычно устанавливается в инструкции ресурса окна. Это должно предотвратить нас от создания этих двух версий вести себя по-другому в этом отношении. Так что версия Справки использует уловку, чтобы делать элемент управление MultiLineEdit только для чтения и фон у нее окрашен в серый цвет, что соответствует только для чтения.

Средство просмотра Справки
Сначала, мы должны обеспечить метод для кнопки pbHelp. Откройте Редактора Окна для DOBDataDialog, выделите кнопку Help и на вкладке General нажмите на свойство Click Event. Измените код следующим образом –
METHOD pbHelp( ) CLASS DOBDataDialog
LOCAL oWin AS HelpViewer
oWin := HelpViewer{SELF}

oWin:Show()
RETURN NIL
Вы возможно теперь предполагаете, что все мы собираемся сделать,  это открыть окно, которое отобразит текст справки. Сохраните вашу работу и закройте Редактор исходного кода и Редактор Окна.

Перед созданием окна HelpViewer, мы создадим основное окно TextViewer, от которого наследуются окна HelpViewer и INIViewer. Только унаследованные версии окна TextViewer будут отображены в приложении.

Создайте новый модуль под называнием "W:TextViewer" с новым объектом "TextViewer" как DialogWindow. Размер Диалогового Окна установите в Ширине 400 и Высоте 310. Добавьте элемент управление MultiLineEdit в позицию 25,36 и размером  350x220. Дайте его свойству Name значение "mleText" и на вкладке Styles, проверьте, что cвойство Return Key установлено в  True для того, чтобы при редактировании текста,  клавиша <Return> могла создовать новую строку (иначе она активизировала бы кнопку OK). Наконец, проверьте, что AutoHScroll установлено в False для того, чтобы текстовые строки в элементе управления MultiLineEdit переносились по словам, чтобы форматировать длине строки.

Добавьте элемент управление PushButton в 160,270 с Caption "OK" и Name "pbOK". Его свойство Click Event здесь не определено, потому что различные версии требуются в каждой унаследованной версии окна.

Сохраните окно и закройте Редактор Окна.

Создайте другой новый модуль "C:HelpViewer". В нем мы подклассифицируем TextViewer как HelpViewer и обеспечим некоторую частность методов к HelpViewer. Выберите модуль C:HelpViewer, откройте Редактор исходного кода и если Вы действительно не хотите набирать  текст программы, импортируйте следующий код из файла Init.prg в каталоге Ch12 на КОМПАКТ-ДИСКЕ (Прим. пер.: Примеры к книге можно скачать с сайта http://ipavels.narod.ru в разделе «Файлы» ) –
METHOD Init(oParent, uExtra) CLASS HelpViewer

LOCAL cString AS STRING

SUPER:Init(oParent, uExtra)

cString := "This application demonstrates the use of " ;

+ "INI files within a class designed for the " ;

+ "purpose. It contains two sections [Min Age] and " ;

+ "[Shell Define]. The file is created on running " ;

+ "the application in C:\TEMP\ with default " ;

+ "contents. " ;

+ CRLF + CRLF ;

+ "The [Min Age] section contains minimum legal " ;

+ "ages for various activities. You could add to " ;

+ "this list outside the application, keeping to " ;

+ "the same format (no spaces either side of the " ;

+ "‘=’ sign). Enter your Date of Birth to check " ;

+ "whether your are 'adult' yet!" ;

+ CRLF + CRLF ;

+ "The [Shell Define] section is updated every " ;

+ "time you leave the application, storing the " ;

+ "current ShellWindow origin, size, and state. " ;

+ "Try changing the size and position or " ;

+ "maximise/minimise the ShellWindow after " ;

+ "closing the Adulthood Checker dialog."
SELF:oDCmleText:Background := Brush{BRUSHLIGHT}

SELF:oDCmleText:TextValue := cString

SELF:oDCmleText:Selection := Selection{0, 0}

SELF:oDCmleText:SetFocus()

SELF:Caption := "Help"
RETURN SELF
Оставьте пустую строку и напечатайте объявление Класса –

CLASS HelpViewer INHERIT TextViewer
Вы увидите, что метод Init() содержит строку, которая будет отображатся как текст справки и свойство MultiLineEdit Backgraund было установлено в серый  цвет (BRUSHLIGHT). Для того чтобы избежать выделения всего текста в элементе управлении, как обычно случается в любом текстовом элементе управлении при первом входе в него, Selection был установлен в Selection {0, 0}, который перемещает курсор перед первым символом. Когда элемент управление MultiLineEdit попадает в фокус, это обеспечивает, что установка Selection вступит в силу.

Нам нужен метод EditFocusChange() для HelpViewer. Создайте скелетный метод на вкладке Control Events для окна TextViewer и когда сделаете сохраняет его, возвратитесь в Репозитарий и переместите объект в модуль C:HelpViewer, нажав на его значок в правой панели Репозитария и перемещая его в модуль C:HelpViewer  в левой панели.

В модуле C:HelpViewer, откройте метод EditFocusChange() и измените имя класса с TextViewer на "HelpViewer" так, чтобы это не имело никакого эффекта на INIViewer и добавьте эти три показанные строки в программу –
METHOD EditFocusChange(oEditFocusChangeEvent) ;

CLASS HelpViewer // Change Name
LOCAL oControl AS Control

LOCAL lGotFocus AS LOGIC
oControl := IIf(oEditFocusChangeEvent == NULL_OBJECT, ;

NULL_OBJECT, ;

oEditFocusChangeEvent:Control)

lGotFocus := IIf(oEditFocusChangeEvent == NULL_OBJECT, ;

FALSE, ;

oEditFocusChangeEvent:GotFocus)

SUPER:EditFocusChange(oEditFocusChangeEvent)

//Put your changes here

IF oControl = SELF:oDCmleText .AND. lGotFocus 
// New Line

SELF:oCCpbOK:SetFocus() 



// New Line

ENDIF






             // New Line
RETURN NIL

Также, добавьте метод pbOK() к C:HelpWiewer –
METHOD pbOK( ) CLASS HelpViewer

SELF:EndDialog(0)

RETURN NIL

Поведение только для чтения сымитировано принудительным фокусом на кнопке OK  каждый раз когда он перемещается на элемент управление MultiLineEdit, так что его там никогда достаточно долго не будет, чтобы можно было изменить текст. Это легко сделано в методе EditFocusChange() как показано выше.

	Рис 12.8
Справочное окно отображает краткое описание приложения. Это Окно унаследовано от окна TextViewer.


Если Вы теперь откомпилируете и запустите приложение, нажав на кнопку Help откроется окно HelpViewer  с текстом справки, показанным на Рис. 12.8.
Средство просмотра INI файла
Теперь о другой версии TextViewer, INIViewer. Сначала создайте метод pbINI() для кнопки pbINI на DOBDialogWindow –
METHOD pbINI( ) CLASS DOBDataDialog
LOCAL oINI AS INIViewer

oWin := INIViewer{SELF}

oWin:Show()
RETURN NIL

Теперь создайте новый модуль с именем "C:INIViewer". Код для INIViewer даже намного проще, чем для HelpViewer –
CLASS INIViewer INHERIT TextViewer
METHOD Init(oParent,uExtra) CLASS INIViewer
LOCAL cString AS STRING

LOCAL dwPosn AS DWORD
SUPER:Init(oParent,uExtra)

cString := MemoRead(oINI:FullPath)

SELF:oDCmleText:TextValue := cString

dwPosn := At("[Shell Define]", cString) - 1

IF dwPosn >= 0

SELF:oDCmleText:Selection := Selection{dwPosn, dwPosn}

SELF:oDCmleText:SetFocus()

ENDIF

SELF:Caption := "Text of " + oINI:FullPath
RETURN SELF
METHOD pbOK( ) CLASS INIViewer
MemoWrit(oINI:FullPath, SELF:oDCmleText:TextValue)

SELF:EndDialog(0)
RETURN NIL
Здесь мы используем <oINI:FullPath>, чтобы возвратить значение C:\TEMP\DOBINI.INI которое используется как параметр для MemoRead(), возвращает содержимое INI файла как строку. Мы устанавливаем MultiLineEdit TextValue в эту строку и затем перемещаем курсор в как раз перед [Shell Define] в случае, если пользователь хочет добавить любые дополнительные элементы в раздел [Min Age]. В pbOK() мы используем MemoWrit(), чтобы записать измененный текст назад в файл.

Откомпилируйте и запустите приложение. Теперь, когда Вы щелкните по кнопке Edit INI, содержание DOBINI.INI файла отобразится как и ожидалось (Рис. 12.9).

	Рис 12.9

Также унаследованное от окна TextViewer, окно просмотра INI файла позволяет нам редактировать содержимое DOBINI.INI файла.


Если Вы отредактировали файл, который заставляет функции WinAPI работать со сбоями, то удалите INI файл и метод App:Start () воссоздаст его во встроенной форме следующий раз при запуске приложения.

Методы со строгим контролем типов

Вы помните, что функции со строгим контролем типов включают два принципа – 
· строгий контроль типов некоторых или всех параметров для того, чтобы несоответствующие параметры нельзя было бы передать в функции, и 

· Строгая типизация возвращаемого значения функции, которая была выполнена объявлена в функциональной инструкции.

Мы можем подобно сжать метод WriteInt(), цитируемый выше, изменяя его первую строку следующим образом –
METHOD WriteInt(cSection AS STRING, cEntry AS STRING, ;

dwINT AS DWORD) AS LOGIC CLASS INIFileSpec
SELF:lFileExists := ;

WritePrivateProfileString(PSZ(cSection), ;

PSZ(cEntry), PSZ(AllTrim(Str(dwINT))), ;

PSZ(SELF:FullPath))
RETURN SELF:lFileExists

Но если мы должны сжать код, как выше, мы должны позволить компилятору узнать об этом Декларированном Классе. Не сделать это приведет к ошибке Missing  declaration  of  typed  method   в Броузере Ошибок.

Добавьте объявление в класс IniFileSpec следующим образом –
CLASS INIFileSpec INHERIT FileSpec

PROTECT lFileExists AS LOGIC

DECLARE METHOD GetInt

DECLARE METHOD GetSection

DECLARE METHOD WriteInt 

// New Line

Вы также помните, что передавали параметр по ссылке (в отличии от массивов и объектов) включая “@” в вызываемый код скорее REF, чем AS рядом с параметром. Он обращается к методам  таким же способом.
STRICT, PASCAL, CLIPPER, и функциях CALLBACK  были обсуждены в Главе 8. Для методов, если неопределенно объявление, CLIPPER назначен по умолчанию, если любые параметры не со строгим контролем типов, в этом случае назначен STRICT. Другие категории не применяются.
Функции Класса
Каждый думает о методах в связи с классами, а не функциями. Однако, есть множество готовых функций, чтобы поддержать ваше использование классов.  Наиболее часто используемые –

· Send() - используется вместо Посылающего Оператора (:). См. Методы CallBack ниже. 
· ASend () - используется, чтобы послать отдельный метод массиву объектов.  

· ClassName () - Часто используется в оценке выражения в сеансе отладки. 

· CreateInstance () - тот же самое что создание объекта через обращение {}. 

· GetAppObject () – Возвращает объект App {}, например посылать ему Сообщение. 

· IsMethod () - Возвращает True, если метод существует для класса, но он не проверяет HIDDEN или PROTECTed методы (см. пример ниже).

· MethodList () - Возвращает массив названий метода для класса, включая методы PROTECTed и HIDDEN. 

· Object2Array () - Возвращает массив, загруженный текущими переменными экземпляра объекта.

Если переменная, содержащая объект со строгим контролем типов и метод посланный этому объекту также со строгим контролем типов, метод будет полностью решен во времени компиляции. Это называется, Ранняя Связывание и означает, что нет никакой информации во время выполнения о методе. Результат этого - то, что некоторые из вышеупомянутых функций не будут иметь информации необходимой для работы должным образом в тех обстоятельствах. Выгоды от раннего связывания (проверка во время компиляции и более быстрое выполнение) перевешивают отказ некоторых  полезных функции. Однако, мы увидим некоторых из этих функций в использовании ниже.

Перегрузка Оператора
Метод Add() является примером метода, используемого для Перегрузки Оператора. Это означает, что, если класс имеет этот метод в своем арсенале, он сможет среагировать на знак «+», как будто был вызван метод Add(). Это - все, что мы будем говорить относительно перегрузки оператора в этой книге.
Защищенные (Protected) и Скрытые (Hidden) Методы

Синтаксис при создании PROTECTed или HIDDEN  метода такой же как и для обычных методов, просто перед первым словом в объявлении метода ставится PROTECTed или HIDDEN слово, как например - PROTECTED METHOD Protected_Method() CLASS Example_Class... и т.д.

Они влияют на видимость метода. Метод HIDDEN - не видим за пределами своего класса, даже в унаследованных классах; точно такое же ограничение как у скрытой переменной экземпляра. Метод PROTECTed видим в унаследованных классах, но не в другом месте в приложения, снова как для переменной экземпляра PROTECTed.

Мы натолкнемся на методы HIDDEN в следующем разделе, где мы захотим запретить пользователю класса использовать методы, предусмотренные только для внутренней работы. Они не скрыты от физического представления в ИНТЕГРИРОВАННОЙ СРЕДЕ РАЗРАБОТКИ (IDE), но появляются в области видимости, когда вызываются снаружи классами.

Рекурсивные Методы
Это было заявлено в Главе 8, в которой функции могли вызываться рекурсивно; тот же самое относится и к методам. Это может быть очень мощный инструмент, хотя иногда трудно понять. Мы можем объединить некоторые идеи выраженные ранее в этой главе, чтобы сохранить массивы со строками разной длины в INI файлам и восстанавливать их как массивы. Только Строки, Дата, Логика и Числа могут  хранится в элементах массива, поскольку мы должны хранить строковые версии данных, это может быть все еще полезным, возможно для отладки или вместо полей МЕМО или BLOB файлов.

Мы напишем рекурсивный код для сохранения следующего строкового массива –
{
Today(),

"London",

TRUE,

124.45,

{ {FALSE, 0.25}, Today(), “New York”} }

к следующему тексту INI файла –

[Array]

.1=D 22/01/99

.2=C London

.3=L .T.

.4=N 124.45

.5=A Array

.5.1=A Array

.5.1.1=L .F.

.5.1.2=N 0.25

.5.2=D 22/01/99

.5.3=C New York

Обратите внимание, что каждый Ключ левые цитаты, адрес элемента массива, и Значение с права показывают тип данных и значение. Например, элемент массива [5,1,2], цитируемый как '.5.1.2 ' является числовым, показанный как ' N ', и значение 0.25. Элементы 5 и 5.1 - самостоятельные массивы и показаны также с их содержанием ниже них.

Это приложение доступно в каталоге Ch12 на CD-ROM как Ch12b. AEF. Однако, весь код дается ниже, и мы предлагаем вам, чтобы Вы работали с нами, т.е. набирали его сами, ручками.

Создайте новое Терминальное Приложение, называв его "Ch12b". Скопируйте модуль C:IniFileSpec из приложения Ch12a в новый, открыв TreeView для Ch12a в Репозитарии и выделив C:IniFileSpec в левой панели. Нажмите <Ctrl-C>. Теперь выделите приложение Ch12b и нажмите <Ctrl-V>. Модуль C:IniFileSpec должен теперь появиться в обоих приложениях.
Откройте окно Application Options и на вкладке Libraries, Вы должны увидеть, что библиотека Terminal Lite присутствует в списке Included; добавьте библиотеку System Classes к этому списку.

Перед внесением изменений в класс IniFileSpec, измените значение по умолчанию в функции Start() -

FUNCTION Start()
LOCAL aArray AS ARRAY

LOCAL oINI AS INIFileSpec
oINI := INIFileSpec{"C:\TEMP\Array.INI"}

aArray := {Today(), "London", TRUE, 124.45, ;

{{FALSE, 0.25}, Today(), "New York"}}

oINI:WriteArray("aArray", aArray)

? MemoRead(oINI:FullPath)

wait

RETURN NIL
В основном, все что  здесь делается это создается Объект IniFileSpec, <oINI>, в который мы записываем содержание <aArray>. Позже мы позже изменим вышеупомянутую функцию Start(), удалив литеральное назначение массива и заменяя его кодом, который будет читать данные массива из INI файла. Естественно, мы должны выполнить вышеупомянутый код сначала так, чтобы INI файл был создан прежде, чем мы пытаемся прочитать его.

Обратите ваше внимания к модуле C: IniFileSpec, мы должны сначала изменить объект для Ккласса IniFileSpec в приложении Ch12b, добавив следующие объявления, потому что методы, которые мы собираемся записать, будут со строгим контролем типов –
DECLARE METHOD Write_Array

DECLARE METHOD WriteArray

DECLARE METHOD WriteString

Выберите C:IniFileSpec, откройте Редактор Исходного кода и введите следующий код, который управляет записью массива в INI файл –
HIDDEN METHOD Write_Array(cArrayName AS STRING, ;

cKeyRoot AS STRING, aArray AS ARRAY) ;

AS LOGIC CLASS INIFileSpec
LOCAL dwCounter, dwLen AS DWORD

LOCAL cType, cValue, cKeyValue AS STRING

LOCAL lSuccess AS LOGIC
dwLen := ALen(aArray)

FOR dwCounter := 1 UPTO dwLen

cType := ValType(aArray[dwCounter])

cValue := Ltrim(AsString(aArray[dwCounter]))

cKeyValue := cType + " "

IF Instr(cType, "CDLN")

cKeyValue += cValue

ELSEIF cType == "A"

cKeyValue += "Array"

ELSE

cKeyValue += "Unknown"

ENDIF

lSuccess := SELF:WriteString(cArrayName, ;

cKeyRoot + "." + NTrim(dwCounter), cKeyValue)

SELF:lFileExists := lSuccess

IF cType == "A" // line 19

lSuccess := SELF:Write_Array(cArrayName, ;

cKeyRoot + "." + NTrim(dwCounter), ;

aArray[dwCounter])

ENDIF

NEXT
RETURN lSuccess

Обратите внимание, что метод Write_Array() вызывает сам себя рекурсивно, то есть если он находит подмассив (как в строке 19), он вызывает себя снова. Типы данных проверяются, используя ValType() который возвращает символ, например 'C' для Строк и ‘A’ для Массивов.

Написав это, мы должны помнить, что Write_Array() не будет удалять никаких  существующих данных этого раздела. Было бы неудобно включить этот процесс в пределах рекурсивно вызываемого метода, но мы можем решить это, представляя новый уровень –
METHOD WriteArray(cArrayName AS STRING, aArray AS ARRAY) ;

AS LOGIC CLASS INIFileSpec
SELF:DeleteSection(cArrayName)
RETURN SELF:Write_Array(cArrayName, NULL_STRING, aArray)

Функция Start() вызывает этот метод, сохраняет <aArray> в раздел с именем 'aArray'. В свою очередь, он вызывает Write_Array() удалявший раздел. WriteArray() нужно только  два параметра, так как мы знаем, что NULL_STRING (определяющий Ключевой корень) не должен быть передан в него. Лучше всего, мы можем сделать метод Write_Array()  HIDDEN, чтобы он появлялся в области видимости когда вызывается снаружи класса, таким образом предотвращая его использование непосредственно. (Вы обратили внимание, что это было определено как HIDDEN выше?)

Мы должны добавить еще два метода к классу IniFileSpec перед проверкой нашей работы - метод DeleteSection() вызвал WriteString(), а метод WriteString() вызвал Write_Array (). Как обычно, откройте Редактора Исходного кода для нового объекта и введите следующий код –
METHOD DeleteSection(cSection) CLASS IniFileSpec

IF SELF:lFileExists

SELF:WriteString(cSection, "", "")

ENDIF

RETURN SELF
METHOD WriteString( cSection AS STRING, ;

cEntry AS STRING, cString AS STRING ) ;

AS LOGIC CLASS INIFileSpec

SELF:lFileExists := ;

WritePrivateProfileString(PSZ(cSection), ;

PSZ(cEntry), PSZ(cString), PSZ( SELF:FullPath))
RETURN SELF:lFileExists

Метод DeleteSection() удаляет раздел INI файла, полученного как параметр и вызывает метод WriteString() записывая индивидуальные данные в INI файл.

Если Вы теперь откомпилируете и выполните приложение, INI файл будет создан и данные массива, будут записаны в него. Функция MemoRead() используется, чтобы отобразить содержимое INI файла в Окне Терминала (См. Рис. 12.10).

	Рис 12.10
Содержимое файла Array.INI, отображенное в Окне Терминала.


Восстановление массива из INI файла, также использует скрытый рекурсивный метод, так что нам нужно объявить два дополнительных метода в определении класса –

DECLARE METHOD GetArray

DECLARE METHOD Get_Array

Чтобы выбрать массив из INI файла, мы должны изменить функцию Start() следующим образом -

FUNCTION Start()
LOCAL aArray AS ARRAY

LOCAL oINI AS INIFileSpec
oINI := INIFileSpec{"C:\TEMP\Array.INI"}

aArray := oINI:GetArray("aArray") // Changed

? MemoRead(oINI:FullPath)

wait
Изменение просто вызывает данные массива, которые нужно прочитать из INI файла в аналогичном, но обратно операции, которая использовалась для записи данных.

Код снова на два метода для того, чтобы 'видимый' метод имел только один параметр, имя Раздела в INI файле. Он работает довольно по-другому, в котором он сначала выбирает целый раздел в <aSection>, который передается в скрытому методу Get_Array(). Снова, нельзя вызывать метод Get_Array () непосредственно в коде, хотя к сожалению ошибка будет замечена только во время выполнения.

Откройте Редактора Исходного кода и введите следующий код –

METHOD GetArray(cArrayName AS STRING) AS ARRAY ;

CLASS INIFileSpec
LOCAL aSection AS ARRAY
aSection := SELF:GetSection(cArrayName)
RETURN SELF:Get_Array(aSection, NULL_STRING)
HIDDEN METHOD Get_Array(aSection AS ARRAY , cKey AS STRING);

AS ARRAY CLASS INIFileSpec
LOCAL cBalance, cType, cValue AS STRING

LOCAL dwCounter AS DWORD

LOCAL aReturn AS ARRAY
aReturn := {}

FOR dwCounter := 1 UPTO ALen(aSection)

cBalance := Stuff(aSection[dwCounter, 1], 1, ;

SLen(cKey) + 1, "")

IF At2(".", cBalance) = 0 .AND. SLen(cBalance) > 0 ;

.AND. Left(aSection[dwCounter, 1], ;

SLen(cKey))== cKey

cType := Left(aSection[dwCounter, 2], 1)

cValue := SubStr2(aSection[dwCounter, 2], 3)
IF cType == "C"

AAdd(aReturn, cValue)

ELSEIF cType == "L"

AAdd(aReturn, cValue == ".T.")

ELSEIF cType == "D"

AAdd(aReturn, CToD(cValue))

ELSEIF cType == "N"

AAdd(aReturn, Val(cValue))

ELSEIF cType == "A"

AAdd(aReturn, SELF:Get_Array(aSection, ;

aSection[dwCounter, 1]))

ELSE

AAdd(aReturn, "Unknown of type " + cType)

ENDIF

ENDIF

NEXT
RETURN aReturn
Пока Вы читаете вышеупомянутое, помните, что < aSection[dwCounter] > выглядит приблизительно –
{".5.1.2", "N 0.25"}

в этом случае < aSection[dwCounter, 2] > будет –
N 0.25

Стоит отметить что функция Stuff() очень полезна при управлении строками. Здесь мы используем ее, чтобы удалить SLen(cKey) + 1 символ с левой стороны aSection[dwCounter, 1] так, чтобы, если <cKey> - '5.1' и < aSection[dwCounter, 1] > - '.5.1.2', он оставляет <cBalance> '2'. (Мы могли одинаково хорошо использовать SubStr2().)

Следующие строки проверяют, что <cBalance> имеет значение без точек и что левая часть < aSection[dwCounter, 1] > пары <cKey> переданы в метод. После этого, мы или добавляем элемент в массив или вызываем метод снова, чтобы иметь дело с подмассивом.

Теперь, когда мы откомпилируем и запустим приложение, единственный источник для данных массива - это INI файл и это все еще указывается, доказывая, что это работало правильно. Когда Вы овладеете отладкой, (Глава 25), Вы также сможете видеть содержание массива в памяти.

Если массив сохраненный в INI файл, имел любые Объекты, Codeblocks (Блоки кода) или другие сложные типы данных, отличные от подмассивов, восстановленный массив будет иметь Unknown of Type O, или подобный.

Когда мы подойдем к отладке в Главе 25, мы увидим  способ, которым можно помещать код этого вида, чтобы хорошо использовать.
Методы Повторного вызова
Метод CallBack - просто метод, для которого обеспечено условие, чтобы вызвать его автоматически в правом точке, если оно существует. Они используются весьма экстенсивно в программировании Windows, вызываемые системными событиями в определенных точках.

Давайте создадим наш собственный, очень простой Метод CallBack. Создайте новое Терминальное Приложение по имени "Ch12c" и введите следующий код в функцию Start() –
FUNCTION Start()
LOCAL oTest AS Test
oTest := Test{}

IF IsMethod(oTest, #Test1)

Send(oTest, #Test1)

ELSE

oTest:Test2()

ENDIF

Wait
RETURN NIL
CLASS Test
/*

METHOD Test1() CLASS Test

? "Method Test1() does exist"

RETURN NIL

*/
METHOD Test2() CLASS Test

? "Method Test1() does NOT exist"

RETURN NIL

В функции Start(), метод Test1() вызывается через функцию Send() (см. Функции Класса выше). Мы использовали функцию Send(), потому что, если бы вызвался нормальный способом, через посылающий оператор, то компилятор выдал бы предупреждение, потому что Test1() не существует (он был закомментирован).

Приложение выполняется без ошибки, потому что функция IsMethod() False и Test1() никогда не вызовется. Если Вы удаляете Операторы Комментария для Test(), он выполнится вместо Test2(). В большинстве случаев, в ваших собственных приложениях, метод или существует и выполняется, или не существует. Мы добавили Test2() чтобы показать то, что случается.

Так что Вы можете создавать метод CallBack (тот, который только выполняется, если он существует)  не вызывая во время работы программы ошибок или предупреждения компилятора, используя функции IsMethod() и Send(), как выше. Но обратите внимание, что компилятор не жаловался на IsMethod(oTest, *Test1) так, очевидно, нет проверки времени компиляции этого функционального вызова. Если бы Вы набрали IsMethod(oTest, *Testt1) он все еще не жаловалось бы, но не выполнило бы таким образом oTest:Test1(). Так что следите за вашим правописанием!

Другой подход состоит в том, чтобы использовать метод NoMethod(). Это - специальный метод, содержание которого Вы должны записать себе и который Вы можете добавлять к вашему классу. Если любое имя метода послано классу, но того метода не существует, то выполнится код в методе NoMethod(). Код мог бы быть что-то вроде этого –
METHOD NoMethod() CLASS Test

ErrorBox{NIL, Symbol2String(NoMethod()) ;

+ " method not defined"}:Show()

RETURN NIL

Функция NoMethod() используемая в пределах этого метода возвратить символическое имя посланного метода.

Вы увидите, что нет необходимости использовать IsMethod() с Send() если Вы включаете в классе метод NoMethod().

Другой подход состоит в том, чтобы использовать CallBack методы NoIVarGet() и NoIVarPut(). Если Вы включаете эти имена методов в ваши новые классы и пишете код для них, то этот код будет работать в следующих обстоятельствах – 

· NoIVarGet() - Если ваш код пробует обращаться к переменной экземпляра, которой не существует, этот метод прерывает запрос. 

· NoIVarPut() - Если ваш код пробует назначить переменной экземпляра , которой не существует, этот метод прерывает запрос.
Вы могли законно спросить: Почему появляются проблемы, при создании ограничений на выполняющийся код с методами, которые не существуют, когда компилятор может указать Вам проблему? По большому счету, Вам будут не нужны эти методы, пока Вы не по пробуете что-то слишком сложное, чтобы писать на этом уровне.
Однако, Вы не хотите включать необходимый код для методов IVarGet(), IVarPut(), и NoMethod() во все ваши классы, по крайней мере не тогда, когда есть более простой путь. Вы можете обойти это, создавая ваш собственный ‘Master Class’ (Главный Класс), который включает их в любой новый класс, в котором Вы нуждаетесь, может наследоваться от этого главного класса, если он не наследуется от какого-нибудь другого класса. Главный Класс мог бы напоминать это –
CLASS MasterClass
METHOD NoIVarGet(symValue) CLASS MasterClass
ErrorBox{, Symbol2String(symValue) ;

+ " IVar not defined"}:Show()
RETURN NIL
METHOD NoIVarPut(symValue) CLASS MasterClass
ErrorBox{, Symbol2String(symValue) ;

+ " IVar not defined"}:Show()
RETURN NIL
METHOD NoMethod() CLASS MasterClass
ErrorBox{, Symbol2String(NoMethod()) ;

+ " method not defined"):Show()
RETURN NIL

Вышеупомянутый класс можно демонстрировать, создаем Empty Application (Постое Приложение) (назовите его "Ch12d") с System и GUI Classes включенными библиотеками. Переименуйте Module1 в " C:Start " и создает новый модуль с именем "C:MasterClass". Откройте Редактора Исходного кода для модуля C:MasterClass и введите код  Декларации Класса и этих трех методов, данных выше.

Перейдите в модуль C:Start, откройте Редактор Исходного кода и вводит следующее –
CLASS SingleClass INHERIT MasterClass
METHOD Start() CLASS App
LOCAL oShell AS ShellWindow

LOCAL aoClass AS ARRAY

LOCAL oClass AS SingleClass

LOCAL uValue AS USUAL
oShell := ShellWindow{SELF}

StatusBar{oShell}

oShell:Show()

aoClass := {}

AAdd(aoClass, SingleClass{})

ASend(aoClass, #Start)

oClass := SingleClass{}

oClass:Name := "EVO"

uValue := oClass:Country

SELF:Exec()
RETURN NIL
SingleClass не имеет никаких методов или IVars, даже нет метода Init(), кроме унаследованных от MasterClass. После показа ShellWindow, единственный Объект SingleClass добавлен к массиву <aoClass>. Функция ASend() (см. Функции Класса выше) посылает методу Start() (не-существующий) элемент, переданный в <aoClass>, но эта строка программы не генерирует предупреждение компилятора, так как не достаточно явно делать так. Вместо этого, ErrorBox (Окно Ошибки) сгенерируется во время выполнения метода NoMethod(), унаследованного от MasterClass.

Метод App:Start() не создает объект <oClass> до окончания показа ошибки NoMethod().

Откомпилируйте и запустите вышеупомянутый код и три сообщения об ошибках, определенные в MasterClass отобразятся как и ожидается (первый из этих сообщений, показаны на Рис. 12.11).

	Рис 12.11
Первое из трех отображенных Диалоговых Ошибок, сгенерированных кодом в классе MasterClass , когда выполнено приложение Ch12d.


Фактически компилятор генерирует предупреждающие сообщения, как показано на Рис. 12.12 (если не установлено, чтобы показать предупреждения как ошибки).

	Рис 12.12
Предупреждающие сообщения, отображаемые в Броузере Ошибок. Они все касаютсясСвойств Country и Name.


Нет никакого предупреждения об использовании метода SingleClass:Start(), обсужденное выше, но предупреждения возникает по упоминанию о свойствах <Name> и <Country>.

Вы в праве спросить, почему используются методы NoIVarGet() и NoIVarPut(), если Вы можете так же просто избегать проблемы, читая предупреждающие сообщения - особенно, если Вы отметили переключатель Показывать Предупреждения как Ошибки в Диалоге Application Options. Первый ответ на это - то, что есть некоторые обстоятельства, где свойственно записать код, который генерирует некоторые предупреждающие сообщения, среди которых Вы не можете определить их сущность. Другой ответ - то, что некоторые классы CA-Visual Objects помещают эти два метода для хорошего использования. Например, класс DBServer Класс использует их, чтобы обеспечить , чтение - запись к обращаемым полям DBF, потому что, во времени компиляции, он не имеет никаких сведений об именах полей. Также, при использовании ActiveX компонентов, Вы можете пробовать обратиться к свойствам и методам, которые не существуют.

Методы IVarGet(), IVarPut() и NoMethod(), конечно же методы CallBack. Они не должны существовать, но если они выполняются, они вызываются с права.

Обычно часто используемый метод CallBack в CA-Visual Objects - Window:EditFocusChange(). Мы использовали его в Приложении Adulthood Checker выше, и он также использовался в Приложении Video. Цитата из системы помощи CA-Visual Objects –
Цель 
Обеспечьте метод, который вызван, когда фокус ввода изменяется на или от средств редактирования. 

Важно! Этот - метод повторного вызова, используемый CA-Visual Objects. Обычно, он не должно вызываться вашим программным код.
Так, если мы создаем метод этого имени для любого класса окна, он будет вызываться системой, каждый раз когда элемент управление получает или теряет фокус (то есть, он  вызывается дважды, интервал между одними элементом редактирования, который теряет фокус и следующий при получении его). Мы создаем метод, но мы не вызываем его; мы оставляем это системе, чтобы она вызвала его в нужное время.

Для того чтобы успешно писать приложения, Вы должны знать то, что Методы Повторного вызова являются доступными по, чтобы найти их в свойствах Окон или Справочном файле. Также необходимо знать, сколько включать в код методов, чтобы они функционировали правильно в сложившейся ситуации, в котором они вызываются. Это сделано относительно простым для нас, как объясняется ниже.

В случае метода EditFocusChange(), мы знаем, что он мог бы содержать его имя и что он делает, но какой код он должен включить, чтобы быть уверенным, работает? CA-Visual Objects содержит ответ по крайней мере в первых нескольких строках. Если Вам нужен этот метод, идите в Редактора Окна и на вкладке Control Events Окна Свойств, Вы найдете свойство EditFocusChange. Выделите его и нажмите на Замещающий знак в справа. Это создаст заготовку метода с необходимым кодом, чтобы дать вам возможность начать -

METHOD EditFocusChange(oEditFocusChangeEvent) ;

CLASS TextViewer
LOCAL oControl AS Control

LOCAL lGotFocus AS LOGIC
oControl := IIf(oEditFocusChangeEvent == NULL_OBJECT, ;

NULL_OBJECT, ;

oEditFocusChangeEvent:Control)

lGotFocus := IIf(oEditFocusChangeEvent == NULL_OBJECT, ;

FALSE, ;

oEditFocusChangeEvent:GotFocus)

SUPER:EditFocusChange(oEditFocusChangeEvent)

//Put your changes here
RETURN NIL

Так, что из вышеупомянутого может быть изменено? В основном, Вы должны включить все вышеописанные линий, как мы делали в окне HelpViewer, выше. Вы найдете, что Вы захотите использовать переменные <oControl> и <lGotFocus>, так что большинство ваших изменений прибудет после того, как им назначат значения. В особенности Вы должны разрешить SUPER вызывать методы, не обязательно сначала, в зависимости от обстоятельств. Вы можете также добавить некоторые локальные переменные для вашего собственного кода, которые, конечно, должны быть помещены в декларацию метода. Но вообще, кроме объявления локальных переменных, ваши изменения будут в области после линии // Put your changes here. Наконец, ваше возвращающий оператор должен быть подобно такой заготовке, чтобы работать с вызвавшим кодом, обычно RETURN NIL. Вернитесь назад к коду, добавленному к этому методу в DOBDataDialog и HelpViewer Классах выше, для примера.

Дело в том , что, если Вы включаете метод этого имени, он будет вызван с права и должен включить код, указанный выше, чтобы работать, но тот код был сделан доступным Вам в понимании способом. Поэтому - никаких проблем! Обратите внимание, что любые изменения, сделанные и сохраненные в Редакторе Окна не имеют никакого эффекта на эти методы, так что они не будут записаны поверх при сохранении изменений дизайна окна.

Большинство Методов CallBack, находятся в состоянии зародыша (подобно вышеупомянутому) работают почти таким же способом. Всегда используйте заготовку, обеспеченные Редактором Окна, чем писать по памяти; это предохранит Вас бесконечных неприятностей.

Стоит упомянуть, что нет никакой связи между вышеупомянутым и объявления функции CallBack в Главе 8. Эти методы не объявлены как CallBack, они только так называются, если они существуют. Можно было предположить, что библиотека GUI Classes содержит большое количество вызовов IsMethod(), он включен в наш код примера выше. Фактически есть фиктивные методы на уровне Класса Окна для большинства этих методов, которые только ни чего не делают и поэтому кажется игнорируются.

Если, из любопытства, Вы решите посмотреть любой из методов CallBack, обеспеченных в пределах Редактора Окна, Вы впоследствии найдете, что все те, которые Вы посмотрели, были созданы. Приложение теперь выполняет незначительно медленнее из-за дополнительно выполняемых методов CallBack. Удостоверитесь, что Вы удалили любые методы, созданные таким образом, если Вы не уверены, что Вы хотите их использовать. Окно Свойств показывает слово 'Defined', когда Вы определили один из этих методов.
Проще и быстрее всего удалить нежелательные методы в Репозитарии, чем в Редакторе Окна.

Методы PreInit () и PostInit () 
Методы Init() классов, созданных в Редакторе Окна и DBServer Редакторе записываются поверх, каждый раз, когда редактор использован и изменения сохранены. Поэтому неблагоразумно изменять код в методах Init(), но есть способ решения этой проблемы.

Первое решение состоит в том, чтобы наследуясь от класса, создать ваш собственный метод Init(), который вызывает SUPER:Init() (как мы делали для DOBShell, который унаследовал от ShellWindow и для HelpViewer и INIViewer, которые оба унаследовали от TextViewer). Другое, более простое решение состоит в том, чтобы поместить ваш собственный код в методы PreInit() и/или PostInit() (как мы делали для DOBDataDialog Окна).

Если Вы создаете ваши собственные методы PreInit() и/или PostInit() для класса, ваш код будет вызываться методом Init() как методы CallBack. Он должен обеспечить всю мощность модификации, в которой Вы будете нуждаться. Методы PreInit() и PostInit() не записываются поверх, когда Вы сохраняете бинарный объект, так что Вы можете безопасно изменять дизайн окна в бинарном редакторе и освежать метод Init() соответственно.

Метод Axit () и функия RegisterAxit ()

Когда Объект прекращает быть связанным с вашим приложением, Сборщик Мусора распорядится им. Типичный пример - то, когда инструкция RETURN метода или функции была достигнута, любой объект, сохраненный в Локальной переменной функции или метод останется в памяти, но без любого 'маркера('дескриптора'), доступного в приложения. Объект, возможно, открыл файл данных и/или занял некоторую память в течение его срока службы, который не имеет никакой значимости после того, как Объект прекращает свое существование. Вы должны убедиться, что закрыли файл или освободили память прежде, чем объект уничтожен, впоследствии не будет способа добраться в эти элементы. Вы можете сделать это явно в некоторых случаях, но проще написать метод Axit(), чтобы сделать это автоматически прежде, чем это – будет слишком поздно. Зарегистрированный  метод Axit() с завершением файла и кодом освобождения памяти будет вызываться Сборщиком Мусора прежде, чем объект уничтожится.

Например, Вы возможно, создали Объект Сервера, который должен быть закрыт прежде, чем его объект будет разрушен. Вы можете создать метод Axit() для любого класса; в этом примере мы создаем для Класса Invoice -

METHOD Axit() CLASS Invoice

SELF:InvoiceServer:Close()

RETURN NIL
Этот метод может содержать код, необходимый для закрытия Сервера или все что еще должно быть сделано; однако, только написать метод - не достаточно. В том смысле что это - не метод CallBack. Он должен быть зарегистрирован, чтобы быть уверенным, что он вызовется. Регистрация достаточно проста –
RegisterAxit(SELF)

Вышеупомянутую строку лучше включить в метод Init() Класса Объекта.

Вы не должны включать и регистрировать метод Axit() для Окон, потому что они имеют метод Close(), который всегда вызывается. Вы должны помнить, что мы использовали этот факт, чтобы сохранить размеры и позицию ShellWindow в INI файл выше.

EXPORT LOCAL
Наконец, если Вы посмотрите на раздел Методы Справки CA-Visual Objects, Вы увидите, что EXPORT LOCAL может быть включен в объявлении метода. Мы оставим это до главы по Блоку кода.

Обзор
В этой главе, мы обсуждали методы, взяв как отправную точку, того что было изучено в Главе 5 (Ой! ОПП) и Глава 8 (Функции и Типы данных).

Мы посмотрели на получении методов устойчивых, в рекурсии и в методах CallBack, PreInit() и PostInit() и Axit().
Мы также упомянули несколько Функций Класса и кратко объяснили использование методов IVarGet(), IVarPut() и NoMethod().

Как полезный класс в своем собственном праве, мы представили IniFileSpec Класс, чтобы обработать INI файлы для простого хранения данных и позже показали, как можно расширить класс, приспособив класс IniFileSpec хранить массив записей разной длины простых элементов типа данных. (Законченный класс IniFileSpec находится в каталоге Ch12 на CD-ROM.)

Мы не пытались объяснить подробно вызовы Windows API, используемые в пределах методов IniFileSpec, потому что они лежат вне области(контекста) этой книги. Однако, этот определенный набор, если скопирован, будет найден полезным и может давать возможность Вам сделать попытку более глубокого изучения Windows API.







